

# Reasoning About Programs

Panagiotis Manolios  
Northeastern University

March 22, 2012

Version: 58

Copyright ©2012 by Panagiotis Manolios

All rights reserved. We hereby grant permission for this publication to be used for personal or classroom use. No part of this publication may be stored in a retrieval system or transmitted in any form or by any means other than for personal or classroom use without the prior written permission of the author. Please contact the author for details.

---

# A Simple Functional Programming Language

In this chapter we introduce a simple functional programming language that forms the core of ACL2s. The language is a dialect of the Lisp programming language and is based on ACL2. In order to *reason* about programs, we first have to understand the *syntax* and *semantics* of the language we are using. The syntax of the language tells us what sequence of glyphs constitute well-formed expressions. The semantics of the language tells us what well-formed expressions (just *expressions* from now on) mean, *i.e.*, how to evaluate them. Our focus is on reasoning about programs, so the programming language we are going to use is designed to be simple, minimal, expressive, and easy to reason about.

What makes ACL2s particularly easy to reason about is the fact that it is a *functional* programming language. What this means is that every built-in function and in fact any ACL2s function a user can define satisfies the rule of Leibniz:

$$\text{If } x_1 = y_1 \text{ and } x_2 = y_2 \text{ and } \dots \text{ and } x_n = y_n, \text{ then } (f\ x_1\ x_2 \dots x_n) = (f\ y_1\ y_2 \dots y_n)$$

Almost no other language satisfies this very strict condition, *e.g.*, in Java you can define a function `foo` of one argument that on input 0 can return 0, or 1, or any integer because it returns the number of times it was called. This is true for Scheme, LISP, C, C++, C#, OCaml, etc. The rule of Leibniz, as we will see later, is what allows us to reason about ACL2s in a way that mimics algebraic reasoning.

You interact with ACL2s via a Read-Eval-Print-Loop (REPL). For example, ACL2s presents you with a prompt indicating that it is ready to accept input.

```
ACL2S BB !>
```

You can now type in an expression, say

```
ACL2S BB !>12
```

ACL2s reads and evaluates the expression and prints the result

```
12
```

It then presents the prompt again, indicating that it is ready for another REPL interaction

```
ACL2S BB !>
```

We recommend that as you read these notes, you also have ACL2s installed and follow along in the “Bare Bones” mode. The “BB” in the prompt indicates that ACL2s is in the “Bare Bones” mode.

The ACL2s programming language allows us to design programs that manipulate objects in the ACL2s *universe*. The set of all objects in the universe will be denoted by `All`. `All` includes:

- ◆ **Rationals:** For example, 11,  $-7$ ,  $3/2$ ,  $-14/15$ .

- ◆ Symbols: For example, `x`, `var`, `lst`, `t`, `nil`.
- ◆ Booleans: There are two Booleans, `t`, denoting *true* and `nil`, denoting *false*.
- ◆ Conses: For example, `(1)`, `(1 2 3)`, `(cons 1 ())`, `(1 (1 2) 3)`.

The Rationals, Symbols, and Conses are disjoint. The Booleans `nil` and `t` are Symbols. Conses are Lists, but there is exactly one list, the empty list, which is not a cons. We will use `()` to denote the empty list, but this is really an abbreviation for the symbol `nil`.

The ACL2s language includes a basic core of built-in functions, which we will use to define new functions.

It turns out that expressions are just a subset of the ACL2s universe. Every expression is an object in the ACL2s universe, but not conversely. As we introduce the syntax of ACL2s, we will both identify what constitutes an expression and what these expressions mean as follows. If *expr* is an expression, then

$$\llbracket expr \rrbracket$$

will denote the semantics of *expr*, or what *expr* evaluates to when submitted to ACL2s at the REPL.

We will introduce the ACL2s programming language by first introducing the syntax and semantics of constants, then Booleans, then numbers, and then conses and lists.

## 1.1 Constants

All constants are expressions. The ACL2s Boolean constant denoting *true* is the symbol `t` and the constant denoting *false* is the symbol `nil`. These two constants are different and they evaluate to themselves.

$$\begin{aligned} \llbracket t \rrbracket &= t \\ \llbracket nil \rrbracket &= nil \\ nil &\neq t \end{aligned}$$

The numeric constants include the natural numbers:

$$0, 1, 2, \dots$$

The negative integers:

$$-1, -2, -3, \dots$$

All integers evaluate to themselves, *e.g.*,

$$\begin{aligned} \llbracket 3 \rrbracket &= 3 \\ \llbracket -12 \rrbracket &= -12 \end{aligned}$$

And the rationals:

$$1/2, -1/2, 1/3, -1/3, 3/2, -3/2, 2/3, -2/3, \dots$$

We will describe the evaluation of rationals in Section 1.3.

## 1.2 Booleans

There are two built-in functions, `if` and `equal`.

When we introduce functions, we specify their *signature*. The signature of `if` is:

$$\text{if} : \text{Boolean} \times \text{All} \times \text{All} \rightarrow \text{All}$$

The signature of `if` tells us that `if` takes three arguments, where the first argument is a `Boolean` and the rest of the arguments are anything at all. It returns anything. So, the signature specifies not only the *arity* of the function (how many arguments it takes) but also its input and output contracts.

Examples of `if` expressions include the following.

```
(if t nil t)
```

```
(if nil 3 4)
```

All function applications in ACL2s are written in prefix form as shown above. For example, instead of `3 + 4`, in ACL2s we write `(+ 3 4)`. The `if` expressions above are elements of the ACL2s universe, *e.g.*, the first `if` expression is a list consisting of the symbols `t`, `nil`, and `t`, in that order.

Not every list starting with the symbol `if` is an expression, *e.g.*, the following are *not* expressions.

```
(if t nil)
```

```
(if 1 3 4)
```

The first list above does not satisfy the signature of `if`, which tells us that the function has an arity of three. The second list also does not satisfy the signature of `if`, which tells us that the input contract requires that the first argument is a `Boolean`. In general, a list is an expression if it satisfies the signature of a built-in or previously defined function.

The semantics of `(if test then else)` is as follows.

$$\llbracket (\text{if } test \text{ then } else) \rrbracket = \llbracket then \rrbracket, \text{ when } \llbracket test \rrbracket = t$$

$$\llbracket (\text{if } test \text{ then } else) \rrbracket = \llbracket else \rrbracket, \text{ when } \llbracket test \rrbracket = \text{nil}$$

For all ACL2s functions we consider, we specify the semantics of the functions only in the case that the signature of the function is satisfied, *i.e.*, only for expressions. If the input contract is violated, then we say that a contract violation has occurred and the function does not evaluate to anything; hence, it does not return a result. For example, as we have seen `(if 1 3 4)` is not an expression. If you try to evaluate it you will get an error message indicating that a contract violation has occurred.

Our first function, `if`, is an important and special function. In contrast to every other function, `if` is evaluated in a *lazy* way by ACL2s. Here is how evaluation works. To evaluate

$$\llbracket (\text{if } test \text{ then } else) \rrbracket$$

ACL2s performs the following steps.

1. First, ACL2s evaluates *test*, *i.e.*, it computes  $\llbracket test \rrbracket$ .

2. If  $\llbracket test \rrbracket = \mathfrak{t}$ , then ACL2s returns  $\llbracket then \rrbracket$ .
3. Otherwise, it returns  $\llbracket else \rrbracket$ .

Notice that *test* is always evaluated, but only one of *then* or *else* is evaluated. In contrast, for all other functions we define, ACL2s will evaluate them in a *strict* way by evaluating all of the arguments to the function and then applying the function to the evaluated results.

Examples of the evaluation of *if* expressions include the following:

$$\begin{aligned} \llbracket (\text{if } \mathfrak{t} \text{ nil } \mathfrak{t}) \rrbracket &= \text{nil} \\ \llbracket (\text{if } \text{nil } 3 \ 4) \rrbracket &= 4 \end{aligned}$$

Here is a more complex *if* expression.

$$(\text{if } (\text{if } \mathfrak{t} \text{ nil } \mathfrak{t}) \ 1 \ 2)$$

This may be confusing because it seems that the test of the *if* is a List, not a Boolean. However, notice that to evaluate an *if*, we evaluate the test first, *i.e.*:

$$\llbracket (\text{if } \mathfrak{t} \text{ nil } \mathfrak{t}) \rrbracket = \text{nil}$$

Therefore,

$$\llbracket (\text{if } (\text{if } \mathfrak{t} \text{ nil } \mathfrak{t}) \ 1 \ 2) \rrbracket = \llbracket 2 \rrbracket = 2$$

The next function we consider is *equal*.

$$\text{equal} : \text{All} \times \text{All} \rightarrow \text{Boolean}$$

$\llbracket (\text{equal } x \ y) \rrbracket$  is  $\mathfrak{t}$  if  $\llbracket x \rrbracket = \llbracket y \rrbracket$  and *nil* otherwise.

Notice that *equal* always evaluates to  $\mathfrak{t}$  or *nil*.

Here are some examples.

$$\begin{aligned} \llbracket (\text{equal } 3 \ \text{nil}) \rrbracket &= \text{nil} \\ \llbracket (\text{equal } 0 \ 0) \rrbracket &= \mathfrak{t} \\ \llbracket (\text{equal } (\text{if } \mathfrak{t} \text{ nil } \mathfrak{t}) \ \text{nil}) \rrbracket &= \mathfrak{t} \end{aligned}$$

That's it for the built-in Booleans constants and functions.

Let us now define some utility functions.

We start with *booleanp*, whose signature is as follows.

$$\text{All} \rightarrow \text{Boolean}$$

The name is the concatenation of the word “*boolean*” with the symbol “*p*.” The “*p*” indicates that the function is a *predicate*, a function that returns  $\mathfrak{t}$  or *nil*. We will use this naming convention in ACL2s (most of the time). Other Lisp dialects indicate predicates using other symbols, *e.g.*, Scheme uses “?” (pronounced “huh”) instead of “*p*.”

Here is how we define functions with contracts in ACL2s.

```
(defunc booleanp (x)
  :input-contract ...
  :output-contract ...)
```

```
(if (equal x t)
    t
    (equal x nil)))
```

The contracts were deliberately elided. We will add them shortly, but first we discuss how to evaluate expressions involving `booleanp`.

How do we evaluate `(booleanp 3)`?

```
[[booleanp 3]]
= { Semantics of booleanp }
  [[(if (equal 3 t) t (equal 3 nil))]]
= { Semantics of equal, [[(equal 3 t)]= nil, Semantics of if }
  [[(equal 3 nil)]]
= { Semantics of equal, [[(equal 3 nil)]= nil }
  nil
```

Above we have a sequence of expressions each of which is equivalent to the next expression in the sequence for the reason given in the hint enclosed in curly braces. For example the first equality holds because we expanded the definition of `booleanp`, replacing the formal parameter `x` with the actual argument `3`.

The next thing is: what is the input contract for `booleanp`?

It is `t` because there are no constraints on the input to the function. All *recognizers* will have an input contract of `t`. A recognizer is a function that given any element of the ACL2s universe recognizes whether it belongs to a particular subset. In the case of `booleanp`, the subset being recognized is the set of Booleans `{t, nil}`.

What about the output contract? Since `booleanp` is a recognizer it returns a Boolean! We express this as follows:

```
(booleanp (booleanp x))
```

So, all together we have:

```
(defunc booleanp (x)
  :input-contract t
  :output-contract (booleanp (booleanp x))
  (if (equal x t)
      t
      (equal x nil)))
```

What does the contract mean? Well, let us consider the general case. Say that function  $f$  with parameters  $x_1, \dots, x_n$  has the input contract  $ic$  and the output contract  $oc$ , then what the contract means is that for any assignment of values from the ACL2s universe to the variables  $x_1, \dots, x_n$ , the following formula is always true.

$$ic \text{ Implies } oc$$

Hence, the contract for `booleanp` means that for any element of the ACL2s universe,  $x$ ,

$$t \text{ Implies } (\text{booleanp } (\text{booleanp } x))$$

If we wanted to make the universal quantification and the implication explicit, we would write the following, where the domain of  $x$  is implicitly understood to be All.

$$\langle \forall x :: t \Rightarrow (\text{booleanp } (\text{booleanp } x)) \rangle$$

Notice that by the relationship between  $\Rightarrow$  (implication) and `if`, the above is equivalent to

$$\langle \forall x :: (\text{if } t \text{ (booleanp (booleanp } x)) \text{ } t) \rangle$$

By the semantics of `if`, we can further simplify this to

$$\langle \forall x :: (\text{booleanp } (\text{booleanp } x)) \rangle$$

So, for any ACL2s element  $x$ , `booleanp` returns a `boolean`.

Let us continue with more basic definitions.

$$\text{and} : \text{Boolean} \times \text{Boolean} \rightarrow \text{Boolean}$$

```
(defunc and (a b)
  :input-contract (if (booleanp a) (booleanp b) nil)
  :output-contract (booleanp (and a b))
  (if a b nil))
```

$$\text{implies} : \text{Boolean} \times \text{Boolean} \rightarrow \text{Boolean}$$

```
(defunc implies (a b)
  :input-contract (and (booleanp a) (booleanp b))
  :output-contract (booleanp (implies a b))
  (if a b t))
```

$$\text{or} : \text{Boolean} \times \text{Boolean} \rightarrow \text{Boolean}$$

```
(defunc or (a b)
  :input-contract (and (booleanp a) (booleanp b))
  :output-contract (booleanp (or a b))
  (if a t b))
```

How do we evaluate the above? Simple:

```
[[or t nil]]
= { Definition of or }
  [[if t t nil]]
= { Semantics of if }
  If [[t]] = nil then [[nil]] else [[t]]
= { Constants evaluate to themselves }
  If t = nil then nil else t
= { t is not nil }
  t
```



**Exercise 1.1** *Define: not, iff, xor, and other Boolean functions.*

`not : Boolean → Boolean`

```
(defunc not (a)
  :input-contract (booleanp a)
  :output-contract (booleanp (not a))
  (if a nil t))
```

`iff : Boolean × Boolean → Boolean`

```
(defunc iff (a b)
  :input-contract (and (booleanp a) (booleanp b))
  :output-contract (booleanp (iff a b))
  (if a b (not b)))
```

`xor : Boolean × Boolean → Boolean`

```
(defunc xor (a b)
  :input-contract (and (booleanp a) (booleanp b))
  :output-contract (booleanp (xor a b))
  (if a (not b) b))
```

### 1.3 Numbers

We have the following built-in recognizers:

`integerp : All → Boolean`

`rationalp : All → Boolean`

Here is what they mean.

`[(integerp x)]` is `t` iff `[x]` is an integer.

`[(rationalp x)]` is `t` iff `[x]` is a rational.

Note that integers are rationals. This is just a statement of mathematical fact.

Notice also that ACL2s includes the *real* rationals and integers, not approximations or bounded numbers, as you might find in most other languages, including C and Java.

We also have the following functions.

`+ : Rational × Rational → Rational`

`* : Rational × Rational → Rational`

`< : Rational × Rational → Boolean`

`unary-- : Rational → Rational`

`unary-/ : Rational → Rational`

Wait, what about (`unary-/ 0`)? The contract really is:

$$\text{unary-}/ : \text{Rational} \setminus \{0\} \rightarrow \text{Rational}$$

How do we express this kind of thing?

```
(defunc unary-/ (a)
  :input-contract (and (rationalp a) (not (equal a 0)))
  ...)
```

The semantics of the above functions should be clear (from elementary school). Here are some examples.

$$\llbracket (+ \ 3/2 \ 17/6) \rrbracket = 13/3$$

$$\llbracket (* \ 3/2 \ 17/6) \rrbracket = 17/4$$

$$\llbracket (< \ 3/2 \ 17/6) \rrbracket = \text{t}$$

$$\llbracket (\text{unary--} \ -2/8) \rrbracket = 1/4$$

$$\llbracket (\text{unary-}/ \ -2/8) \rrbracket = -4$$

**Exercise 1.2** Define subtraction on rationals `-` and division on rationals `/`. Note that the second argument to `/` cannot be 0.

Let's define some more functions, starting with a recognizer for positive integers.

$$\text{posp} : \text{All} \rightarrow \text{Boolean}$$

```
(defunc posp (a)
  :input-contract t
  :output-contract (booleanp (posp a))
  (if (integerp a)
      (< 0 a)
      nil))
```

What if we tried to define `posp` as follows?

```
(defunc posp (a)
  :input-contract t
  :output-contract (booleanp (posp a))
  (and (integerp a)
       (< 0 a)))
```

Well, notice that the contract for `<` is that we give it two rationals. How do we know that `a` is rational? What we would like to do is to test that `a` is an integer first, before testing that `(< 0 a)`, but the only way to do that is to use `if`. This is another reason why `if` is special. When checking the contracts of the `then` branch of an `if`, we can assume that the `test` is *true*; when checking the contracts of an `else` branch, we can assume that the `test` is *false*. No other ACL2s function gives us this capability. If we want to collect together assumptions in order to show that contracts are satisfied, we have to use `if`.

**Exercise 1.3** Define `natp`, a recognizer for natural numbers.

We also have built-in

```
numerator : Rational → Integer
```

```
denominator : Rational → Pos
```

`[(numerator a)]` is the numerator of the number we get after simplifying `[a]`.

`[(denominator a)]` is the denominator of the number we get after simplifying `[a]`.

To simplify an integer `x`, we return `x`.

To simplify a number of the form `x/y`, where `x` is an integer and `y` a natural number, we divide both `x` and `y` by the  $gcd(|x|, y)$  to obtain `a/b`. If `b = 1`, we return `a`; otherwise we return `a/b`. Note that `b` (the denominator) is always positive.

Since rational numbers can be represented in many ways, `ACL2s` returns the simplest representation, *e.g.*,

$$[[2/4]] = 1/2$$

$$[[4/2]] = 2$$

$$[[132/765]] = 44/255$$

## 1.4 Other Atoms

Symbols and numbers are *atoms*. The `ACL2s` universe includes other atoms, such as strings and characters. We'll introduce them later, as needed.

## 1.5 Lists

The only way to create non-atomic data is to use lists.

Our first built-in function is a recognizer for *conses*.

```
consp : All → Boolean
```

Conses are non-empty lists and are comprised of a first element and the rest of the list. Here are the functions for accessing the first and rest of a cons.

```
first : Cons → All
```

```
rest : Cons → All
```

We now define `listp`, a recognizer for lists, as follows.

```
listp : All → Boolean
```

```
(defunc listp (l)
  :input-contract t
  :output-contract (booleanp (listp l))
  (if (consp l)
      (listp (rest l))
```

(equal 1 ( ) ))

The last built-in function is:

$\text{cons} : \text{All} \times \text{List} \rightarrow \text{Cons}$

The semantics of the built-in functions is given by the following rules. Notice that the second argument to `cons` can either be `()` or a `cons`.

$$\begin{aligned} \llbracket (\text{cons } x \text{ ( )}) \rrbracket &= (\llbracket x \rrbracket) \\ \llbracket (\text{cons } x \ y) \rrbracket &= (\llbracket x \rrbracket \ \dots) \text{ where } \llbracket y \rrbracket = (\dots) \\ \llbracket (\text{consp } x) \rrbracket &= \text{t iff } \llbracket x \rrbracket \text{ is a cons.} \end{aligned}$$

Notice that since `consp` is a recognizer it returns a Boolean. So, if  $\llbracket x \rrbracket$  is an atom, then  $\llbracket (\text{consp } x) \rrbracket = \text{nil}$ .

Here are some examples.

$$\begin{aligned} \llbracket (\text{consp } 3) \rrbracket &= \text{nil} \\ \llbracket (\text{consp } (\text{cons } \text{nil } \text{nil})) \rrbracket &= \text{t} \\ \llbracket (\text{consp } \text{nil}) \rrbracket &= \text{nil} \end{aligned}$$

The semantics of `first` and `rest` is given with the following rules.

$$\begin{aligned} \llbracket (\text{first } x) \rrbracket &= a, \text{ where } \llbracket x \rrbracket = (a \ \dots) \text{ for some } a, \dots \\ \llbracket (\text{rest } x) \rrbracket &= (\dots), \text{ where } \llbracket x \rrbracket = (a \ \dots) \text{ for some } a, \dots \end{aligned}$$

Here are some examples.

$$\begin{aligned} \llbracket (\text{first } (\text{cons } (\text{if } \text{t } 3 \ 4) (\text{cons } 1 \text{ ( ) }))) \rrbracket &= 3 \\ \llbracket (\text{first } (\text{rest } (\text{cons } (\text{if } \text{t } 3 \ 4) (\text{cons } 1 \text{ ( ) })))) \rrbracket &= 1 \\ \llbracket (\text{rest } (\text{cons } (\text{if } \text{t } 3 \ 4) (\text{cons } 1 (\text{if } \text{t } \text{nil } \text{t})))) \rrbracket &= (\text{cons } 1 \text{ ( )}) \end{aligned}$$

If you try evaluating `(rest (cons (if t 3 4) (cons 1 (if t () t))))` at the ACL2s command prompt, here is what ACL2s reports.

(1)

Since lists are so prevalent, ACL2s includes a special way of constructing them. Here is an example.

(list 1)

is just a shorthand for `(cons 1 ())`, *e.g.*, notice that asking ACL2s to evaluate

(equal (LIST 1) (cons 1 ()))

results in `t`. What is `list` really? (By the way notice that symbols in ACL2s, such as `list`, are case-insensitive.) It is not a function. Rather, it is a *macro*. There is a lot to say about macros, but for our purposes, all we need to know is that a macro gives us abbreviation power. In general

(list  $x_1$   $x_2$   $\dots$   $x_n$ )

abbreviates (or is shorthand for)

(cons  $x_1$  (cons  $x_2$   $\dots$  (cons  $x_n$  nil)  $\dots$ ))

## 1.6 Contract Violations

Consider

```
(unary-/ 0)
```

If you try evaluating this, you get an error because you violated the contract of `unary-/`. When a function is called on arguments that violate the input contract, we say that the function call resulted in an *input contract violation*. If such a contract violation occurs, then the function does not return anything.

Contract checking is more subtle than this, *e.g.*, consider the following definition.

```
(defunc foo (a)
  :input-contract (integerp a)
  :output-contract (booleanp (foo a))
  (if (posp a)
      (foo (- a 1))
      (rest a)))
```

ACL2s will not admit this function unless it can prove that every function call in the body of `foo` satisfies its contract, a process we call *body contract checking* and that `foo` satisfies its contract, a process we call *contract checking*. This yields five body contract conjectures and one contract conjecture.

**Exercise 1.4** *Identify all the body contract checks and contract checks that the definition of `foo` gives rise to. Which (if any) of these conjectures is always true? Which (if any) of these conjectures is sometimes false?*

Notice that contract checking happens even before the function is admitted. This is called “static” checking. Another option would have been to perform this check “dynamically.” That is, all the contract checking above would be performed as the code is running.

## 1.7 Termination

All ACL2s function definitions have to terminate on all inputs that satisfy the input contract.

For example, consider the following “definition.”

```
(defunc listp (a)
  :input-contract t
  :output-contract (booleanp (listp a))
  (if (consp a)
      (listp a)
      (equal a nil)))
```

ACL2s will not accept the above definition and will report that it could not prove termination.

Let’s look at another example.

Define a function that given  $n$ , returns  $0 + \dots + n$ .

Here are some possibilities:

```
;; sum-n: integer -> integer
```

```
;; Given integer n, return 0+1+2+...+n
(defun sum-n (n)
  :input-contract (integerp n)
  :output-contract (integerp (sum-n n))
  (if (equal n 0)
      0
      (+ n (sum-n (- n 1)))))
(check= (sum-n 5) (+ 1 2 3 4 5))
(check= (sum-n 0) 0)
(check= (sum-n 3) 6)
```

**Exercise 1.5** *The above function does not terminate. Why? Change only the input contract so that it does terminate. Next, change the output contract so that it gives us more information about the type of values `sum-n` returns.*